



# **Side-by-Side Comparison of an eCQM for Eligible Professionals and Eligible Clinicians Using Clinical Quality Language (CQL)**

**Tuesday, December 19, 2017  
1:00-2:00 PM ET**

---

**Shanna Hartman**  
**Centers for Medicare & Medicaid Services**  
**and**  
**Bryn Rhodes, ESAC, Inc.**

# Agenda

- Welcome and Background
- What is Clinical Quality Language (CQL)?
- Measure Tour – CMS 68
- Measure Tour – CMS 124
- Available Tools and Resources


# Background

- CQL is a Health Level Seven International (HL7) standard and aims to unify the expression of logic for electronic clinical quality measures (eCQMs) and Clinical Decision Support (CDS).
- CQL provides the ability to better express logic defining measure populations to improve the accuracy and clarity of eCQMs.
- Benefits of CQL:
  - Improved expressivity
  - More precise/unambiguous
  - Can share logic between measures
  - Can share logic with decision support
  - Can be used with multiple information data models (e.g., Quality Data Model [QDM], Fast Healthcare Interoperability Resources [FHIR])
  - Simplifies calculation engine implementation

# CQL Transition

- eCQMs will be transitioned to use the CQL standard for logic expression
- The transition will begin with the CY 2019 reporting period for Eligible Hospitals and Critical Access Hospitals (CAHs), and CY 2019 performance period for Eligible Professionals (EPs) and Eligible Clinicians for the following programs:
  - Hospital Inpatient Quality Reporting Program
  - Medicare Electronic Health Record Incentive Program for Eligible Hospitals and CAHs
  - Medicaid EHR Incentive Program for EPs, Eligible Hospitals, and CAHs
  - Quality Payment Program: The Merit-based Incentive Payment System (MIPS) and Alternative Payment Models
- To support the transition, CMS will publish CQL-based eCQMs in Spring 2018

# Evolving eCQM Standards




**Definitions:**

**HQMF** – Health Quality Measure Format

**CQL** – Clinical Quality Language

**QDM** – Quality Data Model

# Measure Development – Expected Timelines



a – Measure Structure – HQMF  
b – CQL-based HQMF  
c – Measure Report – QRDA Category I  
(individual report), QRDA Category III  
(aggregate report)

---

# WHAT IS CQL?

# Quality Measurement

- What is a quality measure?
  - Quantitative tool to assess performance related to a specific clinical process or outcome
- What is an electronic Clinical Quality Measure (eCQM)?
  - Electronic representation of a quality measure with the goal of enabling the measure to be evaluated as automatically as possible



# CMS 68 Draft – Description

Percentage of visits for patients aged 18 years and older for which the eligible professional attests to documenting a list of current medications using all immediate resources available on the date of the encounter. This list must include ALL known prescriptions, over-the-counters, herbals, and vitamin/mineral/dietary (nutritional) supplements AND must contain the medications' name, dosage, frequency, and route of administration.

Questions “about” the description:

- Who said it?
- When did they say it?
- What evidence supports it?
- How should I use it?

Metadata


Data Model

Questions about the content of the description:


- What kinds of “things” does it talk about?
- What do those “things” look like?
- What are the relationships between them?
- What are the criteria that apply to them?

Logic

# eCQM Representation




# QDM Data Types



# Encounter, Performed

- Data elements that meet criteria using this datatype should document that the encounter indicated by the QDM category and its corresponding value set is in progress or has been completed.
- Encounter, Performed has the following attributes:
  - Id
  - Code
  - Relevant Period
  - Admission Source
  - Diagnoses
  - Discharge Disposition
  - Length of Stay
  - Negation Rationale
  - Principal Diagnosis
  - Author Datetime
  - Facility Locations

# Components of Sharing Logic



# What is CQL?

- A standard language for expressing clinical knowledge that is
  - Readable
  - Shareable
  - Computable

---

# MEASURE TOURS

# Measure Tours

- A side-by-side review of the
  - Current eCQM specification using QDM 4.3
  - New eCQM specification using QDM 5.3 and CQL
- Tour of two measures:
  - CMS 68 – Documentation of Medications
  - CMS 124 – Cervical Cancer Screening
- NOTE: These draft CQL eCQM specifications are for informational review only



---

# MEASURE TOUR – CMS 68

# Measure Package

## QDM-Based

- .html
- .xml
- \_\_SimpleXML.xml

## CQL-Based

- \_\_HumanReadable.html
- \_\_eCQM.xml
- \_\_CQL.cql
- \_\_ELM.xml
- \_\_ELM.json

NOTE: File-naming conventions within measure packages are still being finalized, and will be posted to the eCQI Resource Center once available. This listing is only to illustrate the types of files that will be present in CQL-based measure packages, and how they compare with the types of files that were present in QDM-based measure packages.

# Measure Metadata

eCQM Title	Documentation of Current Medications in the Medical Record		
eCQM Identifier (Measure Authoring Tool)	10	eCQM Version number	0.0.013
NQF Number	0419	GUID	442f4f7e-3c22-4641-9bee-0e968cc38ef2
Measurement Period	January 1, 20XX through December 31, 20XX		
Measure Steward	Telligen		
Measure Developer	Telligen		
Endorsed By	National Quality Forum		
Description	Percentage of visits for patients aged 18 years and older for which the eligible professional attests to documenting a list of current medications using all immediate resources available on the date of the encounter. This list must include ALL known prescriptions, over-the-counters, herbals, and vitamin/mineral/dietary (nutritional) supplements AND must contain the medications' name, dosage, frequency and route of administration.		
Copyright	Limited proprietary coding is contained in the measure specifications for convenience. Users of the proprietary code sets should obtain all necessary licenses from the owners of these code sets. Quality Insights of Pennsylvania disclaims all liability for use or accuracy of any Current Procedural Terminology (CPT [R]) or other coding contained in the specifications.  CPT (R) contained in the Measure specifications is copyright 2007- 2016 American Medical Association.  LOINC (R) copyright 2004-2015 [2, 54] Regenstrief Institute, Inc. This material contains SNOMED Clinical Terms (R)		

# Measure Contents

## QDM-Based

- Population Criteria
- Data Criteria (QDM Variables)
- Data Criteria (QDM Data Elements)
- Supplemental Data Elements
- Risk Adjustments Variables

## CQL-Based

- Population Criteria
- Definitions
- Functions
- Terminology
- Data Criteria (QDM Data Elements)
- Supplemental Data Elements
- Risk Adjustments Variables

# Population Criteria

## Population Criteria

- **Initial Population** =
  - AND: Age >= 18 year(s) at: "Measurement Period"
  - AND: "Occurrence A of Encounter, Performed: Medications Encounter Code Set" during "Measurement Period"
- **Denominator** =
  - AND: Initial Population
- **Denominator Exclusions** =
  - None
- **Numerator** =
  - AND: "Procedure, Performed: Current Medications Documented SNMD" during "Occurrence A of Encounter, Performed: Medications Encounter Code Set"
- **Numerator Exclusions** =
  - None
- **Denominator Exceptions** =
  - OR: "Procedure, Performed not done: Medical or Other reason not done" for "Current Medications Documented SNMD" during "Occurrence A of Encounter, Performed: Medications Encounter Code Set"
- **Stratification** =
  - None

## Population Criteria

- ▲ **Initial Population**

"Encounters during Measurement Period" Encounter  
where AgeInYearsAt (start of "Measurement Period") >= 18
- ▲ **Denominator**

"Initial Population"
- ▲ **Denominator Exclusions**

None
- ▲ **Numerator**

"Medications Documented"
- ▲ **Numerator Exclusions**

None
- ▲ **Denominator Exceptions**

"Encounters during Measurement Period" Enc  
with "Medications Not Documented for Medical Reason" Meds  
such that Meds.authorDatetime during Enc.relevantPeriod
- ▲ **Stratification**

None

# CQL Library

```
1  library DocumentationofCurrentMedicationsintheMedicalRecord version '0.0.013'  
2  
3  using QDM version '5.3'  
4  
5  valueset "ONC Administrative Sex": 'urn:oid:2.16.840.1.113762.1.4.1'  
6  valueset "Race": 'urn:oid:2.16.840.1.114222.4.11.836'  
7  valueset "Ethnicity": 'urn:oid:2.16.840.1.114222.4.11.837'  
8  valueset "Payer": 'urn:oid:2.16.840.1.114222.4.11.3591'  
9  valueset "Medications Encounter Code Set": 'urn:oid:2.16.840.1.113883.3.600.1.1834'  
10 valueset "Current Medications Documented SNMD": 'urn:oid:2.16.840.1.113883.3.600.1.462'  
11 valueset "Medical or Other reason not done": 'urn:oid:2.16.840.1.113883.3.600.1.1502'  
12  
13 parameter "Measurement Period" Interval<DateTime>  
14  
15 context Patient  
16
```

# Initial Population

## QDM-Based

- **Initial Population =**
  - AND: Age >= 18 year(s) at: "Measurement Period"
  - AND: "Occurrence A of Encounter, Performed: Medications Encounter Code Set" during "Measurement Period"

## CQL-Based

### ▲ Initial Population

"Encounters during Measurement Period" Encounter  
where AgeInYearsAt (start of "Measurement Period") >= 18

# CQL Expressions

- Logic
  - A and B
  - A and not (B or C)
- Comparison
  - A >= B
  - A <> B
- Arithmetic
  - A + B
  - A + B \* C



# Encounters during Measurement

## QDM-Based

- AND: "Occurrence A of Encounter, Performed: Medications Encounter Code Set" during "Measurement Period"

## CQL-Based

### ▲ Encounters during Measurement Period

["Encounter, Performed": "Medications Encounter Code Set"] Encounter  
where Encounter.relevantPeriod during "Measurement Period"

In CQL, specific occurrences are no longer required. Instead, we define an expression that identifies the specific occurrence and use that definition in subsequent criteria.

# Retrieve (square brackets)



- The “type” portion must be the name of some type defined by the model
  - QDM version 5.3 in this case
  - Encounter is the QDM “category”, Encounter, Performed is the datatype
- The “terminology” portion must be a valueset, code, or codesystem
- Result is the set of data elements of the specified *type* that have a *code* that matches the terminology

# Encounters during Measurement

## ▲ Encounters during Measurement Period

`["Encounter, Performed": "Medications Encounter Code Set"] Encounter  
where Encounter.relevantPeriod during "Measurement Period"`

### QDM 4.3:

◦ `AND: "Occurrence A of Encounter, Performed: Medications Encounter Code Set"  
during "Measurement Period"`

- The result of the retrieve is then a *set* of encounters, as opposed to a “yes/no”
- Sets are combined with “intersect” and “union”
  - vs criteria, which are combined with “and” and “or”
- This is a *query*, which is introduced with the “Encounter” *alias*
- The where clause can then use this alias to talk about each encounter in the result

# Filtering with Where

`where Encounter.relevantPeriod during "Measurement Period"`

- “Encounter” refers to the encounters in the “source” of the query
  - In this case, encounters with a code in the appropriate value set
- “Encounter, Performed” structure (i.e. the attributes) is defined by QDM 5.3
  - In this case, the datatype defines attributes such as “relevantPeriod”
- “relevantPeriod” and “Measurement Period” are both DateTime *intervals*
  - Meaning they have a start and end point that is a DateTime value
- CQL supports interval comparisons like this directly
  - e.g. “A during B”, “A overlaps B”, or “A includes B”
- CQL also supports timing phrases
  - e.g. “A starts before start B” or “A starts 1 day or less after end B”

# Timing Relationships

Comparing two date/time values

```
Encounter.authorDatetime < assessment.authorDatetime
```

Comparing a date/time value with an interval (period)

```
assessment.authorDatetime during Encounter.relevantPeriod
```

Comparing an interval with a date/time value

```
Encounter.relevantPeriod includes assessment.authorDatetime
```

Comparing two intervals

```
Encounter.relevantPeriod during "Measurement Period"
```

# Timing and Intervals in CQL

- Full set from QDM
  - starts before start, starts same day as
- Timing phrases
  - starts 3 days before start
  - starts 3 days or less before start
  - starts within 3 days of start
- Interval operators
  - meets, overlaps, during
- Boundary access
  - start of MeasurementPeriod
- Membership
  - X in interval[4, 6]

# Numerator

## QDM-Based

- **Numerator =**
  - AND: "Procedure, Performed: Current Medications Documented SNMD" during "Occurrence A of Encounter, Performed: Medications Encounter Code Set"

## CQL-Based

### ▲ Numerator

"Medications Documented"

### ▲ Medications Documented

"Encounters during Measurement Period" Encounter  
with ["Procedure, Performed": "Current Medications Documented SNMD"] Procedure  
such that Procedure.relevantPeriod during Encounter.relevantPeriod

# Denominator Exceptions

## QDM-Based

- **Denominator Exceptions =**
  - OR: "Procedure, Performed not done: Medical or Other reason not done" for "Current Medications Documented SNMD" during "Occurrence A of Encounter, Performed: Medications Encounter Code Set"

## CQL-Based

### ▲ Denominator Exceptions

"Encounters during Measurement Period" Enc  
with "Medications Not Documented for Medical Reason" Meds  
such that Meds.authorDatetime during Enc.relevantPeriod

### ▲ Medications Not Documented for Medical Reason

["Procedure, Not Performed": "Current Medications Documented SNMD"] Performed  
where Performed.negationRationale in "Medical or Other reason not done"



# Data Criteria

## QDM-Based

- "Encounter, Performed: Medications Encounter Code Set" using "Medications Encounter Code Set Grouping Value Set (2.16.840.1.113883.3.600.1.1834)"
- "Procedure, Performed: Current Medications Documented SNMD" using "Current Medications Documented SNMD SNOMEDCT Value Set (2.16.840.1.113883.3.600.1.462)"
- "Procedure, Performed not done: Medical or Other reason not done" using "Medical or Other reason not done SNOMEDCT Value Set (2.16.840.1.113883.3.600.1.1502)"

## CQL-Based

- "Encounter, Performed: Medications Encounter Code Set" using "Medications Encounter Code Set (2.16.840.1.113883.3.600.1.1834)"
- "Patient Characteristic Ethnicity: Ethnicity" using "Ethnicity (2.16.840.1.114222.4.11.837)"
- "Patient Characteristic Payer: Payer" using "Payer (2.16.840.1.114222.4.11.3591)"
- "Patient Characteristic Race: Race" using "Race (2.16.840.1.114222.4.11.836)"
- "Patient Characteristic Sex: ONC Administrative Sex" using "ONC Administrative Sex (2.16.840.1.113762.1.4.1)"
- "Procedure, Not Performed: Current Medications Documented SNMD" using "Current Medications Documented SNMD (2.16.840.1.113883.3.600.1.462)"
- "Procedure, Performed: Current Medications Documented SNMD" using "Current Medications Documented SNMD (2.16.840.1.113883.3.600.1.462)"

# Terminology

## CQL-Based (only)

### Terminology

- valueset "Current Medications Documented SNMD" using "2.16.840.1.113883.3.600.1.462"
  - valueset "Ethnicity" using "2.16.840.1.114222.4.11.837"
  - valueset "Medical or Other reason not done" using "2.16.840.1.113883.3.600.1.1502"
  - valueset "Medications Encounter Code Set" using "2.16.840.1.113883.3.600.1.1834"
  - valueset "ONC Administrative Sex" using "2.16.840.1.113762.1.4.1"
  - valueset "Payer" using "2.16.840.1.114222.4.11.3591"
  - valueset "Race" using "2.16.840.1.114222.4.11.836"
- 
- Includes all terminologies referenced by the measure
  - This may include direct-reference codes now, rather than only value sets

# Supplemental Data

## QDM-Based

- "Patient Characteristic Ethnicity: Ethnicity" using "Ethnicity CDCREC Value Set (2.16.840.1.114222.4.11.837)"
- "Patient Characteristic Payer: Payer" using "Payer SOP Value Set (2.16.840.1.114222.4.11.3591)"
- "Patient Characteristic Race: Race" using "Race CDCREC Value Set (2.16.840.1.114222.4.11.836)"
- "Patient Characteristic Sex: ONC Administrative Sex" using "ONC Administrative Sex AdministrativeGender Value Set (2.16.840.1.113762.1.4.1)"

## CQL-Based

### ▲ SDE Ethnicity

["Patient Characteristic Ethnicity": "Ethnicity"]

### ▲ SDE Payer

["Patient Characteristic Payer": "Payer"]

### ▲ SDE Race

["Patient Characteristic Race": "Race"]

### ▲ SDE Sex

["Patient Characteristic Sex": "ONC Administrative Sex"]

---

# MEASURE TOUR – CMS 124

# Patient- vs Encounter-based

## Patient-based (CMS124)

Percentage of women 21-64 years of age who were screened for cervical cancer using either of the following criteria:

- \* Women age 21-64 who had cervical cytology performed every 3 years
- \* Women age 30-64 who had cervical cytology/human papillomavirus (HPV) co-testing performed every 5 years.

## Encounter-based (CMS68)

Percentage of visits for patients aged 18 years and older for which the eligible professional attests to documenting a list of current medications using all immediate resources available on the date of the encounter. This list must include ALL known prescriptions, over-the-counters, herbals, and vitamin/mineral/dietary (nutritional) supplements AND must contain the medications' name, dosage, frequency and route of administration.

# Initial Population

## QDM-Based

- AND: Age >= 23 year(s) at: "Measurement Period"
- AND: Age < 64 year(s) at: "Measurement Period"
- AND: "Patient Characteristic Sex: Female"
- AND: Union of:
  - "Encounter, Performed: Office Visit"
  - "Encounter, Performed: Face-to-Face Interaction"
  - "Encounter, Performed: Preventive Care Services - Established Office Visit, 18 and Up"
  - "Encounter, Performed: Preventive Care Services-Initial Office Visit, 18 and Up"
  - "Encounter, Performed: Home Healthcare Services"
  - during "Measurement Period"

# Initial Population

## CQL-Based

### Initial Population

"InDemographics"  
and exists "ValidEncounters"

### InDemographics

AgeInYearsAt(start of "Measurement Period") >= 23  
and AgeInYearsAt(start of "Measurement Period") < 64  
and exists ( ["Patient Characteristic Sex": "Female"] )

### ValidEncounters

( ["Encounter, Performed": "Face-to-Face Interaction"]  
union ["Encounter, Performed": "Office Visit"]  
union ["Encounter, Performed": "Preventive Care Services - Established Office Visit, 18 and Up"]  
union ["Encounter, Performed": "Preventive Care Services-Initial Office Visit, 18 and Up"]  
union ["Encounter, Performed": "Home Healthcare Services"] ) Encounter  
where Encounter.relevantPeriod during "Measurement Period"

# Denominator Exclusions

## QDM-Based

- OR: "Encounter, Performed: Encounter Inpatient (discharge status: Discharged to Home for Hospice Care)" ends during "Measurement Period"
- OR: "Encounter, Performed: Encounter Inpatient (discharge status: Discharged to Health Care Facility for Hospice Care)" ends during "Measurement Period"
- OR: Union of:
  - "Intervention, Order: Hospice care ambulatory"
  - "Intervention, Performed: Hospice care ambulatory"
  - overlaps "Measurement Period"
- OR: "Procedure, Performed: Hysterectomy with No Residual Cervix" ends before end of "Measurement Period"



# Denominator Exclusions

## CQL-Based

### Denominator Exclusions

DenomExcl."Hospice Exclusions"  
or exists "Hysterectomy Procedure"

### DenomExcl.Hospice Exclusions

```
exists ( ["Encounter, Performed": "Encounter Inpatient"] DischargeHospice
  where ( DischargeHospice.dischargeDisposition in "Discharged to Home for Hospice Care"
    or DischargeHospice.dischargeDisposition in "Discharged to Health Care Facility for Hospice Care"
  )
  and DischargeHospice.relevantPeriod ends during "Measurement Period"
)
or exists ( ["Intervention, Order": "Hospice care ambulatory"] HospiceOrder
  where HospiceOrder.authorDatetime during "Measurement Period"
)
or exists ( ["Intervention, Performed": "Hospice care ambulatory"] HospicePerformed
  where HospicePerformed.relevantPeriod overlaps "Measurement Period"
)
```

### Hysterectomy Procedure

```
["Procedure, Performed": "Hysterectomy with No Residual Cervix"] H
where H.relevantPeriod ends before end "Measurement Period"
```

# Measure Libraries

- CQL allows definitions to be shared among measures using Libraries
- Hospice is a library defining hospice exclusions that is shared among multiple measures
- Measure packages will include artifacts for any libraries they reference

# Numerator

## QDM-Based

- AND:
  - OR: "Laboratory Test, Performed: Pap Test (result)" < 3 year(s) ends before end of "Measurement Period"
  - OR:
    - AND: Age >= 30 year(s) at: "Occurrence A of Laboratory Test, Performed: Pap Test"
    - AND: "Occurrence A of Laboratory Test, Performed: Pap Test" satisfies all:
      - (result) < 5 year(s) ends before end of "Measurement Period"
      - satisfies any:
        - <= 1 day(s) starts after or concurrent with start of "Laboratory Test, Performed: HPV Test (result)"
        - <= 1 day(s) starts before start of "Laboratory Test, Performed: HPV Test (result)"

# Numerator

## CQL-Based

### ▲ Numerator

exists "Pap Test Within 3 Years"  
or exists "Pap Test With HPV Within 5 Years"

### ▲ Pap Test Within 3 Years

"Pap Test with Results" Test  
where Test.relevantPeriod 3 years or less before end of "Measurement Period"

### ▲ Pap Test With HPV Within 5 Years

"Pap Test with Results" Test  
with "HPV Test With Results" HPVTest  
such that HPVTest.relevantPeriod starts within 1 day of start of Test.relevantPeriod  
where ( AgeInYearsAt(start of Test.relevantPeriod) >= 30 )  
and Test.relevantPeriod 5 years or less before end of "Measurement Period"

### ▲ Pap Test with Results

["Laboratory Test, Performed": "Pap Test"] LabTest  
where LabTest.result is not null

### ▲ HPV Test With Results

["Laboratory Test, Performed": "HPV Test"] LabTest  
where LabTest.result is not null

# Available Tools and Resources

- CQL Specification - CQL Release 1, Standard for Trial Use (STU) 2
  - [http://www.hl7.org/implement/standards/product\\_brief.cfm?product\\_id=400](http://www.hl7.org/implement/standards/product_brief.cfm?product_id=400)
- CQL-Based HQMF IG – Release 1, STU 2.1
  - [http://www.hl7.org/implement/standards/product\\_brief.cfm?product\\_id=405](http://www.hl7.org/implement/standards/product_brief.cfm?product_id=405)
- eCQI Resource Center
  - CQL Space, including the QDM v5.3 and v5.3 Annotated
    - <https://ecqi.healthit.gov/cql>
  - Check the eCQI Resource Center Events page and CQL Educational Resources page for more information
    - <https://ecqi.healthit.gov/ecqi/ecqi-events>
    - <https://ecqi.healthit.gov/cql/cql-educational-resources>



# Available Tools and Resources (Cont'd)

- CQL Formatting and Usage Wiki
  - <https://github.com/esacinc/CQL-Formatting-and-Usage-Wiki/wiki>
- CQL GitHub Tools Repository
  - [https://github.com/cqframework/clinical\\_quality\\_language](https://github.com/cqframework/clinical_quality_language)
- Measure Authoring Tool
  - <https://www.emeasuretool.cms.gov/>
- Bonnie Testing Tool
  - <https://bonnie.healthit.gov/>
- To submit an issues ticket for CQL, please visit the ONC JIRA site
  - <https://oncprojecttracking.healthit.gov/support/projects/CQLIT>

# Questions?

